

r

The CDF-7/9 filter pair

Construction

Starting with the Daubechies polynomial

```
Daub[N_, z_] := Sum[Binomial[N + k, k] z^k, {k, 0, N}]
```

```
Daub[3, z]
```

```
1 + 4 z + 10 z^2 + 20 z^3
```

Determining the complex roots of this polynomial

```
roots = NSolve[Daub[3, z] == 0, z]
```

```
{{z -> -0.342384}, {z -> -0.078808 - 0.373931 i}, {z -> -0.078808 + 0.373931 i}}
```

There is one real root and two complex conjugate roots. So the Daubechies polynomial can be factored into two polynomials with real coefficients: a linear polynomial $p(z)$ and a quadratic polynomial $q(z)$. The parameter a (which is the leading coefficient of the polynomial $p(z)$) will be fixed later.

```
p[z_] = a (z - (z /. roots[[1]]))
```

```
a (0.342384 + z)
```

```
q[z_] = (20/a) (z - (z /. roots[[2]])) (z - (z /. roots[[3]]))
```

```
 $\frac{1}{a} 20 ((0.078808 - 0.373931 i) + z) ((0.078808 + 0.373931 i) + z)$ 
```

```
Expand[%]
```

```
 $\frac{2.9207 + 0. i}{a} + \frac{(3.15232 + 0. i) z}{a} + \frac{20 z^2}{a}$ 
```

```
Chop[%]
```

```
 $\frac{2.9207}{a} + \frac{3.15232 z}{a} + \frac{20 z^2}{a}$ 
```

The frequency representation of the H-filter that goes with the factor $p(z)$

```
H[w_] = Sqrt[2] Cos[w/2]^4 p[Sin[w/2]^2]
```

```
 $\sqrt{2} a \text{Cos}\left[\frac{w}{2}\right]^4 \left(0.342384 + \text{Sin}\left[\frac{w}{2}\right]^2\right)$ 
```

The frequency representation of the K-filter that goes with the factor $q(z)$

```
K[w_] = Chop[Expand[Sqrt[2] Cos[w/2]^4 q[Sin[w/2]^2]]]
```

```
 $\frac{4.13049 \text{Cos}\left[\frac{w}{2}\right]^4}{a} + \frac{1}{a} 4.45805 \text{Cos}\left[\frac{w}{2}\right]^4 \text{Sin}\left[\frac{w}{2}\right]^2 + \frac{1}{a} 20 \sqrt{2} \text{Cos}\left[\frac{w}{2}\right]^4 \text{Sin}\left[\frac{w}{2}\right]^4$ 
```

Evaluating $H(0)$ (which should be $\sqrt{2}$)

```
H[0]
```

```
0.484204 a
```

```
const = Solve[{H[0] == Sqrt[2]}, {a}]
{{a -> 2.9207}}
```

The Fourier series if the H-filter after fixing the constant

```
H[w_] = H[w] /. const[[1]]
4.13049 Cos[w/2]^4 (0.342384 + Sin[w/2]^2)
```

The Fourier series of the K-filter after fixing the constant

```
K[w_] = K[w] /. const[[1]]
1.41421 Cos[w/2]^4 + 1.52637 Cos[w/2]^4 Sin[w/2]^2 + 9.68408 Cos[w/2]^4 Sin[w/2]^4
```

Checking that all is correct

```
K[0]
1.41421
```

Checking orthogonality conditions for the CDF-7/9 filter pair

```
H[w] Conjugate[K[w]] + H[w + Pi] Conjugate[K[w + Pi]]
4.13049 Cos[w/2]^4 (Conjugate[1.52637 Cos[w/2]^4 Sin[w/2]^2 + 9.68408 Cos[w/2]^4 Sin[w/2]^4] +
  1.41421 Cos[Conjugate[w]/2]^4) (0.342384 + Sin[w/2]^2) + 4.13049 Cos[(pi+w)/2]^4
  (Conjugate[1.52637 Cos[(pi+w)/2]^4 Sin[(pi+w)/2]^2 + 9.68408 Cos[(pi+w)/2]^4 Sin[(pi+w)/2]^4] +
  1.41421 Cos[1/2 (pi + Conjugate[w])]^4) (0.342384 + Sin[(pi+w)/2]^2)
```

```
Assuming[w ∈ Reals, Simplify[%]]
```

```
4.13049 Cos[w/2]^4 (0.342384 + Sin[w/2]^2)
  (1.41421 Cos[w/2]^4 + (0.605255 + 0.0953979 Csc[w/2]^2) Sin[w]^4) +
  4.13049 (0.342384 + Cos[w/2]^2) Sin[w/2]^4
  (1.41421 Sin[w/2]^4 + (0.605255 + 0.0953979 Sec[w/2]^2) Sin[w]^4)
```

```
FullSimplify[Expand[%]]
```

```
2. + 3.33067 × 10-16 Cos[2 w] + 2.77556 × 10-17 Cos[3 w] -
  2.08167 × 10-17 Cos[4 w] - 3.46945 × 10-18 Cos[5 w]
```

```
Chop[%]
```

```
2.
```

Properties of the CDF-7/9 filter pair

```
TrigToExp[H[w]]
```

```
0.788486 + 0.418092 e-i w + 0.418092 ei w -
  0.0406894 e-2 i w - 0.0406894 e2 i w - 0.0645389 e-3 i w - 0.0645389 e3 i w
```

```
hcoeffs = CoefficientList[Expand[Exp[3 I ω %], E^{I ω}]
{-0.0645389, -0.0406894, 0.418092, 0.788486, 0.418092, -0.0406894, -0.0645389}
```

The filter coefficients of the H-filter of length 7

```
hfilter = Table[{k - 4, hcoeffs[[k]]}, {k, 1, 7}];
TableForm[hfilter, TableAlignments -> "."]
```

-3	-0.0645389
-2	-0.0406894
-1	0.418092
0	0.788486
1	0.418092
2	-0.0406894
3	-0.0645389

```
TrigToExp[K[ω]]
```

```
0.852699 + 0.377403 e^{-i ω} + 0.377403 e^{i ω} - 0.110624 e^{-2 i ω} - 0.110624 e^{2 i ω} -
0.0238495 e^{-3 i ω} - 0.0238495 e^{3 i ω} + 0.0378285 e^{-4 i ω} + 0.0378285 e^{4 i ω}
```

```
kcoeffs = Re[CoefficientList[Expand[Exp[4 I ω %], E^{I ω}]]
```


```
{0.0378285, -0.0238495, -0.110624, 0.377403,
0.852699, 0.377403, -0.110624, -0.0238495, 0.0378285}
```

The filter coefficients of the K-filter of length 9

```
kfilter = Table[{k - 5, kcoeffs[[k]]}, {k, 1, 9}];
TableForm[kfilter, TableAlignments -> "."]
```

-4	0.0378285
-3	-0.0238495
-2	-0.110624
-1	0.377403
0	0.852699
1	0.377403
2	-0.110624
3	-0.0238495
4	0.0378285

```
Plot[{H[ω], K[ω]}, {ω, 0, Pi}, PlotStyle -> {Red, Blue}, PlotLabels ->
Placed[{Style["H[ω]", Red, 16], Style["K[ω]", Blue, 16]}, {Above, Above}]]
```



Procedures for the cascade algorithm

```


cascadephidis[h_, n_] :=
Module[{degrees, min, max, sum, hmodif, hpol, hpols, tbl, clist, z},
  (*Approximation of the scaling function belonging
  to the filter h by using the cascade algorithm*)
  degrees = Map[First[#] &, h];
  min = Min[degrees];
  max = Max[degrees];
  sum = Last[Total[h]];
  hmodif = Map[{First[#], Last[#]/sum} &, h];
  hpol := Apply[Plus, Map[Last[#] z^First[#] &, hmodif]];
  hpols := Expand[Product[hpol /. z -> z^(2^t), {t, 0, n-1}]];
  clist = 2^n CoefficientList[z^(-min (2^n-1)) hpols, z];
  tbl = Table[{(k-1) 2^(-n), clist[[k]]}, {k, 1, Length[clist]}];
  ListPlot[tbl, PlotRange -> All, Filling -> Axis, PlotStyle -> PointSize[0.015]]
]

cascadepsidis[h_, n_] :=
Module[{degrees, min, max, sum, hmodif, gpol, hpol, hpols, clist, tbl, z},
  (*Approximation of the wavelet function belonging
  to the filter h by using the cascade algorithm*)
  degrees = Map[First[#] &, h];
  min = Min[degrees];
  max = Max[degrees];
  sum = Last[Total[h]];
  hmodif = Map[{First[#], Last[#]/sum} &, h];
  hpol := Apply[Plus, Map[Last[#] z^First[#] &, hmodif]];
  gpol := hpol /. (z -> -1/z);
  hpols :=
  Expand[Product[hpol /. z -> z^(2^t), {t, 0, n-2}] (gpol /. z -> z^(2^(n-1)))]];
  clist = 2^n CoefficientList[z^(-min (2^(n-1) - 1) + 2^(n-1) max) hpols, z];
  tbl = Table[{(k-1) 2^(-n), clist[[k]]}, {k, 1, Length[clist]}];
  ListPlot[tbl, PlotRange -> All, Filling -> Axis, PlotStyle -> PointSize[0.015]]
]

```


The scaling function belonging to the H-filter of length 7

`cascadephidis[hfilter, 4]`




The wavelet function belonging to the H-filter of length 7

`cascadepsidis[hfilter, 4]`




The scaling function belonging to the K-filter of length 9

`cascadephidis[kfilter, 4]`



The wavelet function belonging to the K-filter of length 9

`cascadepsidis[kfilter, 4]`



Low-pass conditions for the H-filter

$\{H'[\omega], H'[\omega] / . \omega \rightarrow \text{Pi}\}$

$$\{4.13049 \cos\left[\frac{\omega}{2}\right]^5 \sin\left[\frac{\omega}{2}\right] - 8.26098 \cos\left[\frac{\omega}{2}\right]^3 \sin\left[\frac{\omega}{2}\right] \left(0.342384 + \sin\left[\frac{\omega}{2}\right]^2\right), 0.\}$$

$\{H''[\omega], H''[\omega] / . \omega \rightarrow \text{Pi}\}$

$$\{2.06524 \cos\left[\frac{\omega}{2}\right]^6 - 18.5872 \cos\left[\frac{\omega}{2}\right]^4 \sin\left[\frac{\omega}{2}\right]^2 - 4.13049 \cos\left[\frac{\omega}{2}\right]^4 \left(0.342384 + \sin\left[\frac{\omega}{2}\right]^2\right) + 12.3915 \cos\left[\frac{\omega}{2}\right]^2 \sin\left[\frac{\omega}{2}\right]^2 \left(0.342384 + \sin\left[\frac{\omega}{2}\right]^2\right), 0.\}$$

$\{H'''[\omega], H'''[\omega] / . \omega \rightarrow \text{Pi}\}$

$$\{-28.9134 \cos\left[\frac{\omega}{2}\right]^5 \sin\left[\frac{\omega}{2}\right] + 49.5659 \cos\left[\frac{\omega}{2}\right]^3 \sin\left[\frac{\omega}{2}\right]^3 + 20.6524 \cos\left[\frac{\omega}{2}\right]^3 \sin\left[\frac{\omega}{2}\right] \left(0.342384 + \sin\left[\frac{\omega}{2}\right]^2\right) - 12.3915 \cos\left[\frac{\omega}{2}\right] \sin\left[\frac{\omega}{2}\right]^3 \left(0.342384 + \sin\left[\frac{\omega}{2}\right]^2\right), 0.\}$$

$\{H''''[\omega], H''''[\omega] / . \omega \rightarrow \text{Pi}\}$

$$\{-14.4567 \cos\left[\frac{\omega}{2}\right]^6 + 167.285 \cos\left[\frac{\omega}{2}\right]^4 \sin\left[\frac{\omega}{2}\right]^2 - 86.7403 \cos\left[\frac{\omega}{2}\right]^2 \sin\left[\frac{\omega}{2}\right]^4 + 10.3262 \cos\left[\frac{\omega}{2}\right]^4 \left(0.342384 + \sin\left[\frac{\omega}{2}\right]^2\right) - 49.5659 \cos\left[\frac{\omega}{2}\right]^2 \sin\left[\frac{\omega}{2}\right]^2 \left(0.342384 + \sin\left[\frac{\omega}{2}\right]^2\right) + 6.19573 \sin\left[\frac{\omega}{2}\right]^4 \left(0.342384 + \sin\left[\frac{\omega}{2}\right]^2\right), 8.31705\}$$

So the H-filter of length 7 has 4 vanishing moments

Low-pass conditions for the K-filter

$\{K'[\omega], K'[\omega] / . \omega \rightarrow \text{Pi}\}$

$$\{-2.82843 \cos\left[\frac{\omega}{2}\right]^3 \sin\left[\frac{\omega}{2}\right] + 1.52637 \cos\left[\frac{\omega}{2}\right]^5 \sin\left[\frac{\omega}{2}\right] - 3.05273 \cos\left[\frac{\omega}{2}\right]^3 \sin\left[\frac{\omega}{2}\right]^3 + 19.3682 \cos\left[\frac{\omega}{2}\right]^5 \sin\left[\frac{\omega}{2}\right]^3 - 19.3682 \cos\left[\frac{\omega}{2}\right]^3 \sin\left[\frac{\omega}{2}\right]^5, 0.\}$$

$\{K''[\omega], K''[\omega] / . \omega \rightarrow \text{Pi}\}$

$$\{-1.41421 \cos\left[\frac{\omega}{2}\right]^4 + 0.763183 \cos\left[\frac{\omega}{2}\right]^6 + 4.24264 \cos\left[\frac{\omega}{2}\right]^2 \sin\left[\frac{\omega}{2}\right]^2 - 8.39501 \cos\left[\frac{\omega}{2}\right]^4 \sin\left[\frac{\omega}{2}\right]^2 + 29.0523 \cos\left[\frac{\omega}{2}\right]^6 \sin\left[\frac{\omega}{2}\right]^2 + 4.5791 \cos\left[\frac{\omega}{2}\right]^2 \sin\left[\frac{\omega}{2}\right]^4 - 96.8408 \cos\left[\frac{\omega}{2}\right]^4 \sin\left[\frac{\omega}{2}\right]^4 + 29.0523 \cos\left[\frac{\omega}{2}\right]^2 \sin\left[\frac{\omega}{2}\right]^6, 0.\}$$

$\{K'''[\omega], K'''[\omega] / . \omega \rightarrow \text{Pi}\}$

$$\{7.07107 \cos\left[\frac{\omega}{2}\right]^3 \sin\left[\frac{\omega}{2}\right] - 10.6846 \cos\left[\frac{\omega}{2}\right]^5 \sin\left[\frac{\omega}{2}\right] + 29.0523 \cos\left[\frac{\omega}{2}\right]^7 \sin\left[\frac{\omega}{2}\right] - 4.24264 \cos\left[\frac{\omega}{2}\right] \sin\left[\frac{\omega}{2}\right]^3 + 25.9482 \cos\left[\frac{\omega}{2}\right]^3 \sin\left[\frac{\omega}{2}\right]^3 - 280.838 \cos\left[\frac{\omega}{2}\right]^5 \sin\left[\frac{\omega}{2}\right]^3 - 4.5791 \cos\left[\frac{\omega}{2}\right] \sin\left[\frac{\omega}{2}\right]^5 + 280.838 \cos\left[\frac{\omega}{2}\right]^3 \sin\left[\frac{\omega}{2}\right]^5 - 29.0523 \cos\left[\frac{\omega}{2}\right] \sin\left[\frac{\omega}{2}\right]^7, 0.\}$$

$\{K''''[\omega], K''''[\omega] / . \omega \rightarrow \text{Pi}\}$

$$\{3.53553 \cos\left[\frac{\omega}{2}\right]^4 - 5.34228 \cos\left[\frac{\omega}{2}\right]^6 + 14.5261 \cos\left[\frac{\omega}{2}\right]^8 - 16.9706 \cos\left[\frac{\omega}{2}\right]^2 \sin\left[\frac{\omega}{2}\right]^2 + 65.6337 \cos\left[\frac{\omega}{2}\right]^4 \sin\left[\frac{\omega}{2}\right]^2 - 522.941 \cos\left[\frac{\omega}{2}\right]^6 \sin\left[\frac{\omega}{2}\right]^2 + 2.12132 \sin\left[\frac{\omega}{2}\right]^4 - 50.3701 \cos\left[\frac{\omega}{2}\right]^2 \sin\left[\frac{\omega}{2}\right]^4 + 1404.19 \cos\left[\frac{\omega}{2}\right]^4 \sin\left[\frac{\omega}{2}\right]^4 + 2.28955 \sin\left[\frac{\omega}{2}\right]^6 - 522.941 \cos\left[\frac{\omega}{2}\right]^2 \sin\left[\frac{\omega}{2}\right]^6 + 14.5261 \sin\left[\frac{\omega}{2}\right]^8, 18.937\}$$

So the K-filter of length 9 has 4 vanishing moments